Problema 735 de triáguloscabri

Problema. *Sean ABC un triángulo. Si es P un punto sobre su circunferencia circunscrita, sean B' y C' las proyecciones de P sobre las rectas CA y AB, respectivamente, y U el punto medio de $B'C'$. Demostrar que el lugar geométrico de U al variar P es una elipse cuyo centro es el punto medio de la mediana trazada por A.*

Propuesto por Ricardo Barroso Campos.

Solución de Francisco Javier García Capitán.

Esta solución se basa en cálculos con coordenadas baricéntricas que se muestran en las páginas siguientes.



Dado el triángulo ABC, sean
M el punto medio de BC.
O el circuncentro de ABC.
L la segunda intersección de la bisectriz AI con la circunferencia circunscrita.
K el punto simétrico de L respecto de O.
N el punto medio de AM.

- El lugar geométrico buscado es la elipse con centro N tal que uno de sus ejes es paralelo y otro es perpendicular a la bisectriz AI y sus longitudes una es igual a MK y otra es igual a ML.
- Los ejes de la elipse pasan por los puntos medios de MK y ML.
- Se tienen las longitudes

$$MK = \frac{a}{2}\cot\frac{A}{2}, \quad ML = \frac{a}{2}\tan\frac{A}{2}.$$

- La elipse es una circunferencia cuando el triángulo es rectángulo en A.
- Recordemos que las rectas $B'C'$, por ser rectas de Simson del triángulo ABC envuelven a la deltoide de Steiner. Puede comprobarse analíticamente que nuestra elipse y, naturalmente, las correspondientes a los otros vértices, son tangentes a la misma deltoide de Steiner.


A
B
C

Problema 735 de triánguloscabri

```
<< Baricentricas`;
```

Consideramos un punto arbitrario sobre la circunferencia circunscrita. Hallamos su triángulo pedal A1B1C1, que es degenerado, y calculamos el punto medio U de B1C1. Queremos hallar el lugar geométrico de U al variar P.

```
ptP = ConjugadoIsogonal[{-1 - t, 1, t}];
{ptA1, ptB1, ptC1} = TrianguloPedal[ptP];
ptU = Medio[ptB1, ptC1]
```

$$\{-(-1+t)\,(a^2\,(-1+t)+(b^2-c^2)\,(1+t)),\ -(1+t)\,(-a^2+c^2+b^2\,(1+2\,t)),\ -(1+t)\,((-a^2+b^2)\,t+c^2\,(2+t))\}$$

Las coordenadas de U son polinomios de segundo grado respecto del parámetro t. Por tanto, U describirá una cónica. Hallamos su ecuación:

```
conicA = Last[ConicaParametricas[ptU, t]]
```

$$a^2\,x\,y-b^2\,x\,y+c^2\,x\,y+a^2\,y^2-b^2\,y^2-3\,c^2\,y^2+a^2\,x\,z+b^2\,x\,z-c^2\,x\,z-2\,a^2\,y\,z+4\,b^2\,y\,z+4\,c^2\,y\,z+a^2\,z^2-3\,b^2\,z^2-c^2\,z^2$$

Comprobamos que el centro de la cónica es el punto medio de la mediana

```
ptZ = CentroConica[conicA]
```

$\{-2, -1, -1\}$

Obtenemos una gráfica de la cónica :

```
GraficaBaricentricas[conicA]
```



Comprobamos que la cónica es siempre una elipse, viendo que el discriminante es siempre negativo :

```
DiscriminanteConica[conicA]
```

$4\,(a-b-c)\,(a+b-c)\,(a-b+c)\,(a+b+c)$

Para hallar los ejes de la elipse, consideremos una recta cualquiera (q + r) x - 2 q y - 2 r z = 0 que pasa por el centro Z, hallamos su diámetro conjugado y obligamos a que sea perpendicular :

```
rtL = {-q - r, 2 q, 2 r};
rtL1 = Factor[Factor[DiametroConjugado[rtL, conicA]]];
perp = First[Factor[SonPerpendiculares[rtL, rtL1]]]
```

$-4\left(a^2-b^2-c^2\right)(3bq-cq+br-3cr)(3bq+cq+br+3cr)$

Así obtenemos los ejes de la elipse

```
{eje1, eje2} = Map[Simplificar, rtL /. Solve[perp == 0, r]]
```

$\{\{b-c, b+3c, -3b-c\}, \{b+c, b-3c, -3b+c\}\}$

```
GraficaBaricentricas[{conicA, eje1.{x, y, z}, eje2.{x, y, z}},
 {ContourStyle → {Red, Green, Orange}}]
```



Para hallar las longitudes de los ejes de la elipse, hallamos los vértices como intersección de los ejes y la elipse, y hallamos la distancia entre ellos.

```
{ptV1, ptV2} = Map[Simplificar,
   {x, y, z} /. Simplify[Solve[{eje1.{x, y, z} == 0, conicA == 0}, {y, z}], x > 0]];
FullSimplify[CuadradoDistancia[ptV1, ptV2]]
```

$$-\frac{a^2(a-b-c)(a+b+c)}{4(a+b-c)(a-b+c)}$$

```
{ptV3, ptV4} = Map[Simplificar,
   {x, y, z} /. Simplify[Solve[{eje2.{x, y, z} == 0, conicA == 0}, {y, z}], x > 0]];
FullSimplify[CuadradoDistancia[ptV3, ptV4]]
```

$$-\frac{a^2(a+b-c)(a-b+c)}{4(a-b-c)(a+b+c)}$$

Vemos que las longitudes de los ejes son de la forma (a/2)Cot[A/2] y (a/2)Tan[A/2], lo que va a permitir una construcción sencilla de la elipse a partir del centro y las longitudes de los ejes.